Validating Input or “All Input is Evil”

In most applications, the user enters information for the application through the user interface. Data validation ensures that all data entered by a user falls within acceptable parameters before proceeding with program execution. Validating user input reduces the chance of an input error and makes your application more robust (or in other words makes it harder for the user to crash your program).
Rule Number 1:

“All input is evil until proven otherwise” or “you should not trust data until the data is validated”.

Rule Number 2:

“Data must be validated as it crosses the boundary between untrusted and trusted environments”.

Rule Number 3:

“When checking input for validity, you should look for valid data and reject everything else”.  This is done for two reasons;

1. There might be more than on valid way to represent the data.

2. You might miss an invalid data pattern.

There are two types of validation available in VB.NET for user input; form-level validation and field-level validation.  The uses of these two methods in combination dramatically increase the robustness of your application. 
Form-level validation verifies data after the user has filled all of the fields. For example, a user might be directed to fill in a name, address, and phone number on a form and then click OK. With form-level validation, all of the fields on the form would be validated when the user clicked OK.
Field-level validation, on the other hand, verifies the data in each field as the field is filled in. For example, if a user fills in a field that holds a phone number, field-level validation could verify that the number contains a valid area code before moving to the next field. As each digit is entered, you could also use control events to verify that only numbers are entered.

Field-level validation

You might want to validate data as it is entered into each field. Field-level validation gives the developer control over user input as it occurs. 
The most commonly used control for user input is the TextBox. Several properties of the TextBox control allow you to restrict the values of user input that they will accept. Some of these properties include:

· MaxLength (limits the number of characters that can be entered into a text box)

· PasswordChar (allows you to hide user input at run time)

· ReadOnly (determines whether a user can edit the value displayed in a text box)

· MultiLine (determines whether a text box can accept multiple lines)
· CharacterCasing (formats the case of the input i.e. Upper or Lower)
Validating Characters 

The Char data type contains several Shared (static) methods that are useful for validating characters trapped by the KeyPress event. These methods include 

· Char.IsDigit 
· Char.IsLetter 

· Char.IsLetterOrDigit 

· Char.IsPunctuation 

· Char.IsLower 

· Char.IsUpper 

Each of these methods evaluates a character and returns a Boolean value. The Char.IsDigit function returns true if a character is a numeric digit, false if it is not. The Char.IsLower function returns true if a character is a lowercase letter, false otherwise. The other methods behave similarly. The following example uses the Char.IsNumber method to test if the key pressed is a numeric key: 

Private Sub TextBox1_KeyPress (ByVal sender as Object, ByVal e As _
   System.Windows.Forms.KeyPressEventArgs) Handles TextBox1.KeyPress
   If Char.IsDigit(e.KeyChar) = True Then
      MessageBox.Show("You pressed a number key")
   End If
End Sub

The Validating and Validated Events 

The easiest way to validate data is to use the Validating event. The Validating event occurs before a control loses the focus. This event is raised only when the CausesValidation property of the control that is about to receive the focus is set to true. Thus, if you want to use the Validating event to validate data entered in your control, the CausesValidation of the next control in the tab order should be set to true. In order to use Validating events, the CausesValidation property of the control to be validated must also be set to true. By default, the CausesValidation property of all controls is set to true when controls are created at design time. Controls such as Help buttons are typically the only kind of controls that have CausesValidation set to false.
The Validating event allows you to perform sophisticated validation on your controls. You could, for example, implement an event handler that tested whether the value entered corresponded to a very specific format. Another possible use is an event handler that doesn't allow the focus to leave the control until a value has been entered.
The Validating event includes an instance of the CancelEventArgs class. This class contains a single property, Cancel. If the input in your control does not fall within required parameters, you can use the Cancel property within your event handler to cancel the Validating event and return the focus to the control.
The Validated event fires after a control has successfully been validated. You can use this event to perform any actions based upon the validated input.
The following example demonstrates a handler for the Validating event. This method requires an entry in TextBox1 before it will allow the focus to move to the next control
Private Sub TextBox1_Validating(ByVal sender As Object, ByVal e As _
   System.ComponentModel.CancelEventArgs) Handles TextBox1.Validating
   ' Checks the value of TextBox1
   If TextBox1.Text = "" Then 
      ' Resets the focus if there is no entry in TextBox1
      e.Cancel = True
   End If
End Sub

To use the Validating event of a text box

1. Add a text box to a form
2. Create an event handler to handle the Validating event of the text box. In the event handler, set the e.Cancel property to true to cancel validating and return the focus to the text box.
3. Set the CausesValidation property to false for any controls for which you do not want the Validating event to fire
Form-level validation

Form-level validation is the process of validating all of the fields on a form at once. A central procedure is used to implement form-level validation and is usually called when the user is ready to proceed to another step.

The following example demonstrates how to create a form-level validation method. The sample tests that all the text boxes on a form have received input when a button called btnValidate is pressed, and resets the focus to the first text box it encounters without input
Private Sub btnValidate_Click(ByVal sender As System.Object, ByVal e _
   As System.EventArgs) Handles btnValidate.Click
   Dim aControl As System.Windows.Forms.Control
   ' Loops through each control on the form
   For Each aControl In Me.Controls
      ' Checks to see if the control being considered is a Textbox and
      ' if it contains an empty string
      If TypeOf aControl Is TextBox AndAlso aControl.Text = "" Then
      ' If a textbox is found to contain an empty string, it is 
         ' given the focus and the method is exited.
         aControl.Focus()
         Exit Sub
      End If
   Next
End Sub

Providing User Feedback

When invalid input is entered into a field, the user should be alerted and given an opportunity to correct the error. 
The ErrorProvider component provides an easy way to communicate validation errors to your users. The ErrorProvider allows you to set an error message for each control on your form when the input is not valid. When an error message is set, an icon indicating the error will appear next to the control, and the error message text will be shown as a Tool Tip when the mouse hovers over the affected control. The ErrorProvider component can be found in the Windows Forms tab of the Toolbox

To create a validation handler that uses the ErrorProvider component 

1. Create your form and add an ErrorProvider component to your form. The ErrorProvider component appears in the component tray. 

2. Set the CausesValidation property of the control you want to provide errors for to true if it is not true already. 

3. In the event handler for that control's Validating event, test the value. If an error condition occurs, use the SetError method to set the error to be displayed. The following example demonstrates a validation handler for a text box named pswordTextBox and an error provider named myErrorProvider:
Private Sub pswordTextBox_Validating(ByVal sender as Object, _
   ByVal e As System.ComponentModel.CancelEventArgs) Handles _
   pswordTextBox.Validating
   ' Validate the entry
   If pswordTextBox.Text = "" Then
      ' Set the error for an invalid entry
      myErrorProvider.SetError(pswordTextBox, _
         "Password cannot be blank!")
   Else
      ' Clear the error for a valid entry-no error will be displayed
      myErrorProvider.SetError(pswordTextBox, "")
   End If
End Sub

Using Regular Expressions for Checking Input:
For simple data validation, you can use code which matches patterns of characters within strings.
Overview

To validate text fields, such as names, addresses, tax identification numbers, and so on, use regular expressions to do the following: 

· Constrain the acceptable range of input characters. 

· Apply formatting rules. For example, pattern-based fields, such as tax identification numbers, ZIP Codes, or postal codes, require specific patterns of input characters. 

· Check lengths. 
Regular expression support is available to .NET applications through the RegularExpressionValidator control and the Regex class in the System.Text.RegularExpressions namespace.

Using a RegularExpressionValidator Control

If you capture input, you can use the RegularExpressionValidator control to validate that input. You can use regular expressions to restrict the range of valid characters, to strip unwanted characters, and to perform length and format checks. You can constrain the input format by defining patterns that the input must match.

To validate a server control's input using a RegularExpressionValidator 

1. Add a RegularExpressionValidator control to your form. 

2. Set the ControlToValidate property to indicate which control to validate. 

3. Set the ValidationExpression property to an appropriate regular expression. 

4. Set the ErrorMessage property to define the message to display if the validation fails. 

The code below demonstrates two functions to validate data (one using a regular expression).
Private Function ValidateName() As Boolean

  Const MAX_NAME_LENGTH = 50

  If txtName.Text.Trim() = "" Then

    ErrorProvider1.SetError(txtName, _

    "You must enter the bowler's name.")

    Return False

  End If

  If txtName.Text.Length > MAX_NAME_LENGTH Then

    ErrorProvider1.SetError(txtName, _

      "The bowler's name is too long.")

  End If

  'A valid name is of the form 

  'FirstName LastName or (Word)(Space)(Word).

  'The last name can contain 

  'symbols such as commas, periods, apostrophes, 

  'dashes and spaces (used in limited ways).

  If Not Regex.IsMatch(txtName.Text, _

      "^[a-zA-Z]+ ([a-zA-Z])?(\')?([a-zA-Z]+)?(\-)?" & _

      "[a-zA-Z]+(\,)?(\ )?([a-zA-Z]+)?(\.)?$") Then

    ErrorProvider1.SetError(txtName, _

      "The bowler's name contains invalid characters.")

    Return False

  End If

  Return True

End Function

Private Function ValidateScore() As Boolean

  If Not IsNumeric(txtScore.Text) OrElse _

      Integer.Parse(txtScore.Text) < 0 OrElse _

      Integer.Parse(txtScore.Text) > 300 Then

        ErrorProvider1.SetError(txtScore, _

          "You must enter a bowling score between 0 and 300.")

    Return False

  End If

  Return True

End Function

Common Regular Expressions
Some common regular expressions are shown in Table 1, and the meta-characters that are used within these expressions are shown and explained in Table 2.
Table 1. Common Regular Expressions

	Field 
	Expression
	Format Samples
	Description

	Name
	^[a-zA-Z''-'\s]{1,40}$
	John Doe
O'Dell
	Validates a name. Allows up to 40 uppercase and lowercase characters and a few special characters that are common to some names. You can modify this list.

	Social Security Number
	^\d{3}-\d{2}-\d{4}$
	111-11-1111
	Validates the format, type, and length of the supplied input field. The input must consist of 3 numeric characters followed by a dash, then 2 numeric characters followed by a dash, and then 4 numeric characters.

	Phone Number
	^[01]?[- .]?(\([2-9]\d{2}\)|[2-9]\d{2})[- .]?\d{3}[- .]?\d{4}$
	(425) 555-0123
425-555-0123
425 555 0123
1-425-555-0123
	Validates a U.S. phone number. It must consist of 3 numeric characters, optionally enclosed in parentheses, followed by a set of 3 numeric characters and then a set of 4 numeric characters. 

	E-mail 
	^([0-9a-zA-Z]([-.\w]*[0-9a-zA-Z])*@([0-9a-zA-Z][-\w]*[0-9a-zA-Z]\.)+[a-zA-Z]{2,9})$
	someone@example.com
	Validates an e-mail address.

	URL
	^(ht|f)tp(s?)\:\/\/[0-9a-zA-Z]([-.\w]*[0-9a-zA-Z])*(:(0-9)*)*(\/?)([a-zA-Z0-9\-\.\?\,\'\/\\\+&amp;%\$#_]*)?$
	http://www.microsoft.com
	Validates a URL 

	ZIP Code
	^(\d{5}-\d{4}|\d{5}|\d{9})$|^([a-zA-Z]\d[a-zA-Z] \d[a-zA-Z]\d)$
	12345
	Validates a U.S. ZIP Code. The code must consist of 5 or 9 numeric characters.

	Password
	(?!^[0-9]*$)(?!^[a-zA-Z]*$)^([a-zA-Z0-9]{8,10})$
	 
	Validates a strong password. It must be between 8 and 10 characters, contain at least one digit and one alphabetic character, and must not contain special characters.

	Non- negative integer
	^\d+$
	0
986
	Validates that the field contains an integer greater than zero.

	Currency (non- negative)
	^\d+(\.\d\d)?$
	1.00
	Validates a positive currency amount. If there is a decimal point, it requires 2 numeric characters after the decimal point. For example, 3.00 is valid but 3.1 is not.

	Currency (positive or negative)
	^(-)?\d+(\.\d\d)?$
	1.20
	Validates for a positive or negative currency amount. If there is a decimal point, it requires 2 numeric characters after the decimal point.

	Date (mm/dd/yy)
	[012]?\d\/[0123] ?\d\/[0]\d
	12/24/02
	Month values can begin with a 0, 1, or 2 if they’re two digits long. Day values can begin with a 0, 1, 2, or 3 if they’re two digits long (such as 12/24/02).

	Date (dd-MMM-yyyy)
	[0-3]\d-(JAN│FEB│ MAR│APR│MAY│JUN│JUL│ AUG│SEP│OCT│NOV│DEC) -\d{4}
	29-JAN-2002
	Matches dates that use one of the proscribed month short forms (such as 29-JAN-2002)


Table 2. Regular expression meta-characters 

	Character
	Rule

	{m}
	Requires m repetitions of the preceding character. For example, 7{3} matches 777.

	{m, n}
	Requires m to n repetitions of the preceding character. For example, 7{2,3} matches 77 and 777 but not 7777.

	*
	Zero or more occurrences of the previous character or subexpression. For example, 7*8 matches 7778 or just 8.

	+
	One or more occurrences of the previous character or subexpression. For example, 7+8 matches 7778 or 78 but not just 8.

	?
	One or zero occurrences of the previous character or subexpression. For example, 7?8 matches 78 and 8 but not 778.

	( )
	Groups a sub expression that will be treated as a single element. For example, (78)+ matches 78 and 787878.

	│
	Either of two matches. For example 8│6 matches 8 or 6.

	[ ]
	Matches one character in a range of valid characters. For example, [A-C] matches A, B, or C.

	[^ ]
	Matches a character that isn’t in the given range. For example, [^A-B] matches any character except A and B.

	.
	Any character except newline.

	\s
	Any whitespace character (such as a tab or space).

	\S
	Any non-whitespace character.

	\d
	Any digit character.

	\D
	Any non-digit character.

	\w
	Any “word” character (letter, number, or underscore).

	\W
	Any non-word character.

	\
	Use to search for a special character. For example, use \\ for the literal \ and use \+ for the literal +.

	^
	Represents the start of the string. For example, ^777 can only find a match if the string begins with 777.

	$
	Represents the end of the string. For example, 777$ can only find a match if the string ends with 777.


Using Parse Method for Checking Input:

The Parse method is a general-purpose method available on most data types that enables you to convert data from one type to any other type. In the process of conversion, the Parse method validates that the source data can be converted to the destination data type.
The following code demonstrates how you can create a general-purpose function that uses the Parse method for validating strings containing formatted monetary amounts. The IsValidMoneyFormat function validates that the monetary string passed in via the strFormattedMoneyAmount parameter is in an acceptable format, and upon validation returns the converted amount as a Decimal value (in a ByRef parameter).

Public Function IsValidMoneyFormat(ByVal strFormattedMoneyAmount As String, _

                                       ByRef DecimalAmount As Decimal) _

                                       As Boolean

  Dim fIsValidFormat As Boolean = True

  DecimalAmount = 0

  Try

    'TODO: Add Imports System.Globalization to the top of the code module

    DecimalAmount = Decimal.Parse(strFormattedMoneyAmount, _

                                    NumberStyles.Currency)

  Catch ex As Exception

    'Take any exception as a signal that the format of the passed 

    'in string is invalid

     fIsValidFormat = False

  End Try

  Return fIsValidFormat

End Function

